Mappe affini

Ver 0.1
Michele Andreoli

Internamente la mappa affine { x'=f(x,y), y'=g(x,y)} verra' rappresentata da una matrice 3x3. Per costruire la rappresentazione interna, si usera' un "costruttore" principale Affinita[], e vari generatori da questo derivati.
La rappresentazione matriciale e' del tipo 3x3:

$$\textbf{Matrice} = \begin{pmatrix} r * \text{Cos}[\varphi] & -s * \text{Sin}[\psi] & a \\ r * \text{Sin}[\varphi] & s * \text{Cos}[\psi] & b \\ 0 & 0 & 1 \end{pmatrix}$$

Queste matrici M agiscono su vettori del tipo (x,y,1) (il terzo valore, puo' essere utile per rappresentare il colore del punto).

Con queste notazioni, r ed s sono le lunghezze dei nuovi versori; phi e' l'angolo del nuovo versore x rispetto all'asse y e psi e' l'angolo del nuovo versore y, rispetto all'asse y. (a,b) e' la traslazione dell'origine.
Per identificare visivamente le matrici rappresentanti le affinita', si usera' il simbolo map[M]: in sostanza, map[] rappresentera' *un' applicazione che associa ad una matrice 3x3 la mappa affine corrispondente.*

■ Costruttori di mappe

Il costruttore base richiede espressamente tutti i parametri dell'affinita'

```
Clear[map, Affinita]
Affinita[ phi_, psi_, r_, s_, a_, b_] :=
 map[ { {r * Cos[phi], -s * Sin[psi], a},
        {r * Sin[phi], s * Cos[psi], b},
        {0, 0, 1}
      }]
```

```
General::spell1 : Possible spelling error: new symbol name "map" is similar to existing symbol "Map".
```

Puo' anche essere utile un costruttore che accetti direttamente l'intera matrice:

```
Affinita[ M_MatrixQ] := map[M]
```

Com'e' ovvio, questo costruttore ritorna direttamente la rappresentazione di M quale mappa affine.
La regola seguente impone a *Mathematica* di stampare ogni oggetto del tipo map[x] nella forma esplicita matriciale, ma cancellando la terza riga, perche' priva di informazioni.

```
Format[ map[x_]] := MatrixForm[Drop[x, {3}]]
```

Esempio:

```
generica = Affinita[phi, psi, r, s, a, b]
```

$$\begin{pmatrix} r\,\text{Cos}[\text{phi}] & -s\,\text{Sin}[\text{psi}] & a \\ r\,\text{Sin}[\text{phi}] & s\,\text{Cos}[\text{psi}] & b \end{pmatrix}$$

■ Operatori standard che agiscono su oggetti di tipo map[]

In *Mathematica* e' possibile ridefinire l'azione di un operatore (anche nativo) su un tipo di oggetto definito dall'utente, come negli esempi che seguono.
Avendo scelto una rappresentazione 3x3, la composizione viene effettuata semplicemente moltiplicando le matrici corrisponenti.

```
map /: Composition[ map[A_], map[B_]] := map[ A . B // Simplify]
```

Con questa speciale sintassi si definisce il modo in cui Composition[] (che e' un costrutto nativo di *Mathematica)* deve agire sugli oggetti di tipo map[]. Altre utili ridefinizioni sono le seguenti:

```
map /: Det[ map[A_]] := Simplify[Det[A]]
map /: Inverse[ map[A_]] := map[Inverse[A] // Simplify]

Det[generica] // Simplify
```

```
r s Cos[phi - psi]
```

```
Inverse[generica]
```

$$\begin{pmatrix} \frac{\text{Cos[psi] Sec[phi-psi]}}{r} & \frac{\text{Sec[phi-psi] Sin[psi]}}{r} & -\frac{\text{Sec[phi-psi] (a Cos[psi]+b Sin[psi])}}{r} \\ -\frac{\text{Sec[phi-psi] Sin[phi]}}{s} & \frac{\text{Cos[phi] Sec[phi-psi]}}{s} & \frac{\text{Sec[phi-psi] (-b Cos[phi]+a Sin[phi])}}{s} \end{pmatrix}$$

■ Generatori speciali

```
Rotazione[alpha_] := Affinita[alpha, alpha, 1, 1, 0, 0]
Translazione[ {a_, b_}] := Affinita[0, 0, 1, 1, a, b]
Stiramento[r_, s_] := Affinita[0, 0, r, s, 0, 0]
Scala[r_] := Affinita[0, 0, r, r, 0, 0]
SimmetriaSpeciale[alpha_] := Affinita[2 * alpha, 2 * alpha, 1, -1, 0, 0]
```

Note: SimmetriaSpeciale[a] e' una simmetria con asse passante per (0,0), formante un angolo "a" con l'asse x.

■ Qualche esempio d'uso dei generatori

```
Clear[a]

rot = Rotazione[a]
```

$$\begin{pmatrix} \text{Cos[a]} & -\text{Sin[a]} & 0 \\ \text{Sin[a]} & \text{Cos[a]} & 0 \end{pmatrix}$$

Osservare che, benche' "rot" venga stampato come una matrice, esso e' internamente rappresentato come un oggetto di tipo map[matrice], e puo' essere usato in ogni posto dove useremmo map[matrice].
Calcoliamo determinante e l'inversa di rot:

```
Det[rot] // Simplify
```

```
1
```

```
Inverse[rot]
```

$$\begin{pmatrix} \text{Cos}[a] & \text{Sin}[a] & 0 \\ -\text{Sin}[a] & \text{Cos}[a] & 0 \end{pmatrix}$$

```
Composition[ rot, Inverse[rot]]
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \end{pmatrix}$$

Composizioni varie:

```
Composition[ Scala[s], rot]
```

$$\begin{pmatrix} s\,\text{Cos}[a] & -s\,\text{Sin}[a] & 0 \\ s\,\text{Sin}[a] & s\,\text{Cos}[a] & 0 \end{pmatrix}$$

```
SimmetriaSpeciale[a]
```

$$\begin{pmatrix} \text{Cos}[2\,a] & \text{Sin}[2\,a] & 0 \\ \text{Sin}[2\,a] & -\text{Cos}[2\,a] & 0 \end{pmatrix}$$

```
Composition[ Translazione[{tx, ty}], Rotazione[a]]
```

$$\begin{pmatrix} \text{Cos}[a] & -\text{Sin}[a] & tx \\ \text{Sin}[a] & \text{Cos}[a] & ty \end{pmatrix}$$

■ Applicazione delle mappe a punti

Poiche' map[] un'applicazione tra lo spazio delle matrici 2x3 e il gruppo delle affinita', questo vuol dire che map[M], dove M e' una matrice, rappresenta effettivamente una trasformazione affine che puo' essere applicata ad un punto (x,y) come faremmo per le altre funzioni di *Mathematica*: {x',y'}=map[M][{x,y}].
Questo fondamentale azione viene definita con la seguente istruzione:

```
map[M_] [ {x_, y_}] := Drop[M . {x, y, 1}, {3}]
map[M_][{x_, y_, z_}] := M.{x, y, z}
```

Esempio di uso:

```
rot[ {x, y}]
```

```
{x Cos[a] - y Sin[a], y Cos[a] + x Sin[a]}
```

■ Applicazione delle mappe ad oggetti geometrici

Molti oggetti geometrici (linee, poligoni, etc) hanno gia' una loro rappresentazione in *Mathematica*. Occorre insegnare a map[] in quale maniera deve agire su di essi. E' chiaro, che questo verra realizzato sfruttando il fatto che map[] sa gia' come agire su un singolo punto.
Un poligono, ad esempio, e' internamente rappresentato da *Mathematica* nella forma Polygon[{ P1, P2, ...}].
Per dire che la mappa m deve agire su tutti i punti del poligono, si scrive:

```
(m_map)[ Polygon[punti_]] := Polygon[ Map[m, punti]]
```

Analogamente per altri oggetti gia' noti a *Mathematica*:

```
(m_map)[Line[punti_]] := Line[Map[m, punti]]
```

Esempio di applicazione:

```
Rotazione[Pi / 2][   Line[ {{x1, y1}, {x2, y2}} ]   ]
```

```
Line[{{-y1, x1}, {-y2, x2}}]
```

Per applicare un'affinita' a tutti i punti di una lista, si e' usato la funzione Map[operatore, lista] di *Mathematica*.

■ Esempi

Mathematica 4 supporta internamente la creazione di parecchi oggetti geometri fondamentali. Occorre caricare il seguente package:

```
Needs["Geometry`Polytopes`"]
```

Se ora volessi i vertici dell'ottagono regolare, basta scrivere:

```
oct0 = Polygon[Vertices[Octagon]]
```

$$\text{Polygon}\left[\left\{\left\{\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right\}, \{0, 1\}, \left\{-\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right\}, \{-1, 0\}, \left\{-\frac{1}{\sqrt{2}}, -\frac{1}{\sqrt{2}}\right\}, \{0, -1\}, \left\{\frac{1}{\sqrt{2}}, -\frac{1}{\sqrt{2}}\right\}, \{1, 0\}\right\}\right]$$

Se ora volessi applicare la seguente affinita':

```
aff = Composition[ Translazione[{3, 0}], Rotazione[0.1], Stiramento[1, 2]]
```

$$\begin{pmatrix} 0.995004 & -0.199667 & 3. \\ 0.0998334 & 1.99001 & 0. \end{pmatrix}$$

devo fare:

```
oct1 = aff[ oct0]
```

```
Polygon[
 {{3.56239, 1.47774}, {2.80033, 1.99001}, {2.15524, 1.33656}, {2.005, -0.0998334},
  {2.43761, -1.47774}, {3.19967, -1.99001}, {3.84476, -1.33656}, {3.995, 0.0998334}}]
```

```
Show[ Graphics[ oct0], Graphics[oct1],
 AspectRatio → Automatic, Frame → True, GridLines → Automatic]
```




```
- Graphics -
```

■ Rappresentazione del reticolo trasformato

```
Clear[ret]; reticolo = {};
Do[ AppendTo[reticolo, {i, j}], {i, 1, 10}, {j, 1, 10}]

aff = Composition[Translazione[{3, 0}], Rotazione[Pi / 8], Stiramento[3, 1]]
ListPlot[Map[aff, reticolo], AspectRatio → Automatic]
```

$$\begin{pmatrix} 3\,\mathrm{Cos}[\frac{\pi}{8}] & -\mathrm{Sin}[\frac{\pi}{8}] & 3 \\ 3\,\mathrm{Sin}[\frac{\pi}{8}] & \mathrm{Cos}[\frac{\pi}{8}] & 0 \end{pmatrix}$$



- Graphics -

■ Manipolazione di immagini fotografiche

```
pink = Import["/root/mm/pink.bmp"]
```

- Graphics -

```
Show[pink]
```



- Graphics -

```
bm = pink[ [1, 1]];
```

```
Dimensions[bm]
```

{34, 71}

```
DensityGraphics[bm] // Show
```

- DensityGraphics -

PARAGRAFO DA SCRIVERE

DA PARAGRAFO SCRIVERE

■ Composizioni di simmetrie

Componendo due simmetrie assiali formanti angolo x, si ottiene una rotazione di 2x

```
Sx = SimmetriaSpeciale[0];
Sy = SimmetriaSpeciale[Pi / 2];
S1 = SimmetriaSpeciale[a];
Composition[ S1, Sx]
```

$$\begin{pmatrix} \text{Cos}[2\,a] & -\text{Sin}[2\,a] & 0 \\ \text{Sin}[2\,a] & \text{Cos}[2\,a] & 0 \end{pmatrix}$$

Componendo le simmetrie assiali ortogonali Sx and Sy, si ottiene la simmetria centrale:

```
Composition[Sx, Sy]
```

$$\begin{pmatrix} -1 & 0 & 0 \\ 0 & -1 & 0 \end{pmatrix}$$

■ Simmetrie assiali in generale

Se proviamo a comporre una simmetria speciale St (retta t che passa per l'origine) con una simmetria Sr (retta r che NON passa per l'origine, di angolo alpha e intercetta q) si ottiene una translazione nel verso che va da (t) ad (r), di lunghezza pari al doppio della distanza tra le due rette t ed r: Sr * St = T(2u), e quindi Sr=T(2u)*St^−1=T(2u)*St

```
Simmetria[ alpha_, q_] := Module[ {u, d},
  d = q * Cos[alpha];
  u = {-Sin[alpha], Cos[alpha]};
  Composition[ Translazione[ u * 2 * d], SimmetriaSpeciale[alpha]]
 ]
```

Pi/2 lo consideriamo un caso a parte:

```
Simmetria[ Pi / 2, q_] := Composition[Translazione[-u * 2 * q], SimmetriaSpeciale[Pi / 2]]
```

Esempi:

```
Simmetria[0, q]
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & -1 & 2\,q \end{pmatrix}$$

```
Simmetria[Pi / 4, q]
```

$$\begin{pmatrix} 0 & 1 & -q \\ 1 & 0 & q \end{pmatrix}$$

```
Simmetria[Pi / 2, q]
```

$$\text{Composition}\left[\text{Translazione}[-2\,q\,u],\ \begin{pmatrix} -1 & 0 & 0 \\ 0 & 1 & 0 \end{pmatrix}\right]$$

```
Simmetria[Pi / 6, q]
```

$$\begin{pmatrix} \frac{1}{2} & \frac{\sqrt{3}}{2} & -\frac{\sqrt{3}\,q}{2} \\ \frac{\sqrt{3}}{2} & -\frac{1}{2} & \frac{3\,q}{2} \end{pmatrix}$$

Componiamo ora due simmetrie ad assi paralleli; dovremmo ottenere una pura translazione:

```
Composition[ Simmetria[alpha, q + k], Simmetria[alpha, q]]  // Simplify
```

$$\begin{pmatrix} 1 & 0 & -k\,\mathrm{Sin}[2\,\mathrm{alpha}] \\ 0 & 1 & 2\,k\,\mathrm{Cos}[\mathrm{alpha}]^2 \end{pmatrix}$$